


Skriv ut primtall i Flash med ActionScript 3.0 (Eratosthenes' sil)

I matematikk og informatikk er en algoritme en presis beskrivelse av en endelig serie operasjoner som skal utføres for å løse et problem eller et sett med flere problem. Her får du oppgitt en algoritme for å finne primtall og skal skrive et program i ActionScript som utfører instruksjonene. Et primtall er et naturlig tall som bare kan deles på 1 og seg selv uten å gi rest. For å finne alle primtall mindre enn eller lik et gitt heltall n kan du bruke Eratosthenes' algoritme:

1. Lag en liste av påfølgende heltall fra **2** til **n**: (2, 3, 4, ..., n).
2. La **p** til å begynne med være lik **2**, det første primtallet.
3. Stryk ut alle multipler av **p** som er større enn **p**, fra listen.
4. Finn det første tallet større enn **p** som står igjen på listen. Dette tallet er det neste primtallet. Sett **p** lik dette tallet.
5. Gjenta trinn **3** og **4** inntil p^2 er større enn **n**.
6. Alle gjenværende tall på listen er primtall.

Se om du får til noe som ligner utskriften nedenfor. Dersom du trenger hjelp, kan du kikke i [vedlegget](#) eller laste ned kildefilen <http://tip.no/skole/res/EratosthenesSil fla>.



Vedlegg (Kildekode)

```

1  /*****
2  * Eratosthenes' sil er i matematikk en enkel, *
3  * eldgammel algoritme for å finne alle primtall *
4  * opp til et spesifikt naturlig tall. *
5  * *
6  * Ola Lie 12.06.12 *
7  *****/
8
9  /*****
10 * 1. Lag en liste av påfølgende heltall *
11 * fra 2 til N: (2, 3, 4, ..., N). *
12 *****/
13 var liste:Array = [];
14 var i:int;
15 const N:int = 1000;
16 for (i=2; i<=N; i++) liste.push(i);
17
18 /*****
19 * 2. La p til å begynne med være lik 2, *
20 * det første primtallet. *
21 *****/
22 var p:int = 2;
23
24 do { // se punkt 5
25
26     /*****
27     * 3. Stryk ut alle multipler av p *
28     * som er større enn p, fra listen. *
29     *****/
30     for (i=liste.length-1; i>0; i--) {
31
32         if (liste[i]%p==0 && liste[i]>p) {
33             liste.splice(i,1);
34         } // slutt if
35
36     } // slutt for
37
38
39     /*****
40     * 4. Finn det første tallet større enn p *
41     * som står igjen på listen. *
42     * Dette tallet er det neste primtallet. *
43     * Sett p lik dette tallet. *
44     *****/
45     for (i=0; i<liste.length; i++) {
46
47         if (liste[i]>p) {
48             p = liste[i];
49             break;
50         } // slutt if
51
52     } // slutt for
53
54 /*****
55 * 5. Gjenta trinn 3 og 4 *
56 * inntil p^2 er større enn N. *
57 *****/
58 }
59 while (p*p <= N);
60
61 // trace("Liste",liste);
62

```

Koden for å legge tallene ut på visningsområdet følger på neste side.

```
63 /*****
64  * Skriv ut primtallene
65  * Se http://tip.no/skole/pdf/TabellUtskrift.pdf * TabellUtskrift.pdf
66  *****/
67 const ANTALL_KOLONNER:int = 12;
68 var rad, kol:int;
69 var format:TextFormat = new TextFormat();
70 format.size = 20;
71 format.font = "Bitstream Vera Sans Mono";
72 format.align = TextFormatAlign.RIGHT;
73
74 for (i=0; i<liste.length; i++) {
75     rad = int(i/ANTALL_KOLONNER); // se http://tip.no/skole/it/norskeFlagg.php
76     kol = i%ANTALL_KOLONNER;
77     var tf:TextField = new TextField(); norskeFlagg.php
78     tf.text = liste[i];
79     tf.setTextFormat(format);
80     tf.x = 50 + kol*50;
81     tf.y = 100 + rad*30;
82     addChild(tf);
83 } // slutt ytre for
```